TP n°2 : Graphiques

Exercice 1. Tracer la tangente en un point

Écrire des commandes permettant de tracer la représentation graphique de la fonction exponentielle et sa tangente au point $(0, 1)$ pour des abscisses allant de -1 à 1.

```
--> x=-1:0.01:1; x=x'; plot2d(x,[exp(x),1+x])
```

Exercice 2. Illustration graphique des développements limités

Écrire une ligne de commandes permettant de tracer les courbes des fonctions f et g définies par

$$f(x) = \ln(1 + x) \text{ et } g(x) = x - \frac{x^2}{2}.$$

On fera varier x dans $[-0.999, 4]$.

```
--> x=-0.999:0.01:4; x=x'; plot2d(x,[log(1+x),x-x.^2/2])
```

Exercice 3. Courbe en cloche de Gauss

Écrire une ligne de commandes permettant de tracer la "courbe en cloche" de Gauss, c'est-à-dire la courbe qui représente la densité usuelle $\varphi_{m,\sigma}$ de la loi normale de paramètres m et σ^2. On rappelle que :

$$\forall x \in \mathbb{R}, \quad \varphi_{m,\sigma}(x) = \frac{1}{\sigma\sqrt{2\pi}} e^{\frac{-(x-m)^2}{2\sigma^2}}.$$

On demande sur un même graphique les courbes correspondant à $m = 0$ et $\sigma^2 = 1$, $m = 0$ et $\sigma^2 = 2$, ainsi que $m = 0$ et $\sigma^2 = 3$. On fera varier x dans l'intervalle $[-4, 4]$.

```
--> function y=f(x,sigma), y=exp(-x.^2/(2*sigma^2))/(sigma*sqrt(2*%pi)), endfunction
--> x=-4:0.01:4; x=x'; plot2d(x,[f(x,1),f(x,sqrt(2)),f(x,sqrt(3))])
```

Exercice 4. Tracer la fonction réciproque

Définir une subdivision de l'intervalle $I = [-2, 2]$ en 100 sous-intervalles de même longueur. Écrire une ligne de commandes permettant de tracer sur un même dessin la courbe de la fonction f définie sur I par

$$f(x) = \frac{e^x - e^{-x}}{2},$$

ainsi que celle de sa fonction réciproque (sans chercher à déterminer cette dernière !).

```
--> x=linspace(-2,2,101); y=(exp(x)-exp(-x))/2; plot2d(x,y), plot2d(y,x)
```

En effet, on a l'équivalence $y = f(x) \Leftrightarrow x = f^{-1}(y)$. Ceci montre que si l'on demande y en fonction de x pour tracer la courbe de f, il suffit de demander x en fonction de y pour obtenir celle de f^{-1}.

Exercice 5. Tracer une nappe

1. Écrire des commandes permettant de tracer la nappe représentant le fonction F définie par

$$\forall (x, y) \in \mathbb{R}^2, \quad F(x, y) = \frac{3}{4} + \frac{1}{2}\left(x^2 + 4xy + y^2\right) + x^2y^2.$$

On fera varier (x, y) dans $[-1, 1] \times [-1, 1]$.

```
--> function z=F(x,y), z=3/4+1/2*(x^2+4*x*y+y^2) + x^2*y^2, endfunction
--> x=-1:0.01:1; y=x; fplot3d(x,y,F)
```

2. Écrire des commandes permettant de tracer la nappe représentant le fonction G définie par

$$\forall (x, y) \in \mathbb{R}^2, \quad G(x, y) = x^2\left(1 + y^3\right) + y^2.$$

On fera varier (x, y) dans $[-1, 1] \times [-2, 2]$, puis dans $\left[-\frac{1}{2}, \frac{1}{2}\right] \times \left[-\frac{1}{2}, \frac{1}{2}\right]$.

```
--> function z=G(x,y), z=x^2*(1+y^3) + y^2, endfunction
--> x=-1:0.01:1; y=-2:0.01:2; fplot3d(x,y,G)
```

puis

```
--> x=-1/2:0.01:1/2; y=-1/2:0.01:1/2; fplot3d(x,y,G)
```

3. Écrire des commandes permettant de tracer la nappe représentant le fonction H définie par

$$\forall (x,y) \in \mathbb{R}^2, \quad H(x,y) = xe^{x\left(y^2+1\right)}.$$

On fera varier (x,y) dans $[-3,0] \times [-1,1]$.

```
--> function z=H(x,y), z=x*exp(x*(y^2+1)), endfunction
--> x=-3:0.01:0; y=-1:0.01:1; fplot3d(x,y,H)
```

Exercice 6. Tracer un diagramme circulaire et un histogramme

On considère la commande suivante

```
n=input('entrez la valeur de n'), x=grand(1,n,'bin',10,0.5)
```

On dispose ainsi d'une série statistique X à valeurs entières.

1. Écrire des commandes permettant de tracer le diagramme circulaire associé à cette série.

```
--> n=input('entrez la valeur de n'), x=grand(1,n,'bin',10,0.5)
--> m=tabul(x,'i'); pie(m(:,2))
```

2. Écrire des commandes permettant de tracer un histogramme associé à cette série en 5 classes de même amplitude.

```
--> n=input('entrez la valeur de n'), x=grand(1,n,'bin',10,0.5)
--> c=linspace(0,10,6); histplot(c,x)
```